

# INCLUDE

|  |
|--|
| <b>INCLUDE</b> <i>copycode-name</i> [ <i>operand1 ... 99</i> ] |
|--|

This chapter covers the following topics:

- Function
- Syntax Description
- Examples

For an explanation of the symbols used in the syntax diagram, see *Syntax Symbols*.

---

## Function

The **INCLUDE** statement is used to include source lines from an external object of type copycode into another object at compilation.

The **INCLUDE** statement is evaluated at *compilation* time. The source lines of the copycode will not be physically included in the source of the program that contains the **INCLUDE** statement, but they will be included during the program compilation and thus in the resulting object module.

### Caution:

A source code line which contains an **INCLUDE** statement must not contain any other statement.

## Syntax Description

Operand Definition Table:

| Operand         | Possible Structure | Possible Formats | Referencing Permitted | Dynamic Definition |
|-----------------|--------------------|------------------|-----------------------|--------------------|
| <i>operand1</i> | C                  | A U              | no                    | no                 |

Syntax Element Description:

| Syntax Element       | Description   |
|----------------------|---|
| <i>copycode-name</i> | <p><b>Copycode Name:</b></p> <p>As <i>copycode-name</i> you specify the name of the copycode whose source is to be included.</p> <p><i>copycode-name</i> may contain an ampersand (&amp;); at compile time, this character will be replaced by the one-character code corresponding to the current value of the Natural system variable *LANGUAGE. This feature allows the use of multilingual copycode names.</p> <p>The object you specify must be of the type copycode. The copycode must be contained either in the same library as the program which contains the INCLUDE statement or in the respective steplib (the default steplib is SYSTEM).</p> <p>When the source of a copycode is modified, all programs using that copycode must be compiled again to reflect the changed source in their object codes.</p> <p>The source code of the copycode must consist of syntactically complete statements.</p> |

| Syntax Element  | Description   |
|-----------------|---|
| <i>operand1</i> | <p><b>Insert Values for Dynamic Insertion:</b></p> <p>You can dynamically insert values in the copycode which is included. These values are specified with <i>operand1</i>.</p> <p>In the copycode, the values are referenced with the following notation:</p> <p><code>&amp;n&amp;</code></p> <p>That is, you mark the position where a value is to be inserted with <code>&amp;n&amp;</code>. <i>n</i> is the sequential number of each value passed with the INCLUDE statement. For example, <code>&amp;3&amp;</code> would refer to the third value specified with the statement.</p> <p>For every <code>&amp;n&amp;</code> notation in the copycode you must specify a value in the INCLUDE statement. For example, if the copycode contains <code>&amp;5&amp;</code>, <i>operand1</i> must be specified at least five times.</p> <p>You may write one copy code parameter (<code>&amp;n&amp;</code>) after another without blanks (that is, <code>&amp;1&amp;&amp;2&amp;&amp;3&amp;</code>). This method is used to concatenate multiple copy code parameters to a source.</p> <p>A string may follow one or several copy code parameters without a blank (that is, <code>&amp;1&amp;abc</code> or <code>&amp;1&amp;&amp;2&amp;abc</code>). This method is used to concatenate a string to multiple copy code parameters.</p> <p><b>Note:</b><br/>Because <code>&amp;n&amp;</code> is a valid part of an identifier, this notation may not be used as a copy code parameter substitution in other positions described above (i.e. <code>abc&amp;1&amp;</code> or <code>&amp;1&amp;abc&amp;2&amp;</code>). In other words, a string may only come after copy code parameters, not before or between.</p> <p>Values that are specified in the INCLUDE statement but not referenced in the copycode will be ignored.</p> |

## Examples

- Example 1 - INCLUDE Statement Including Copycode
- Example 2 - INCLUDE Statement Including Copycode with Parameters
- Example 3 - INCLUDE Statement Using Nested Copycodes
- Example 4 - INCLUDE Statement with Concatenated Parameters in Copycode

### Example 1 - INCLUDE Statement Including Copycode

**Program containing the INCLUDE statement:**

```

** Example 'INCEX1': INCLUDE (include copycode)
*****
*
WRITE 'Before copycode'
*
INCLUDE INCEX1C
*
WRITE 'After copycode'
*
END

```

**Copycode INCEX1C to be included:**

```

** Example 'INCEX1C': INCLUDE (copycode used by INCEX1)
*****
*
WRITE 'Inside copycode'

```

**Output of Program INCEX1:**

```

Page          1                                05-01-25  16:26:36

Before copycode
Inside copycode
After copycode

```

**Example 2 - INCLUDE Statement Including Copycode with Parameters****Program INCEX2 containing the INCLUDE statement:**

```

** Example 'INCEX2': INCLUDE (include copycode with parameters)
*****
DEFINE DATA LOCAL
1 EMPL-VIEW VIEW OF EMPLOYEES
  2 NAME
END-DEFINE
*
*
INCLUDE INCEX2C 'EMPL-VIEW' 'NAME' '' 'ARCHER' '' '20' '' 'BAILLET' ''
*
END

```

**Copycode INCEX2C to be included:**

```

** Example 'INCEX2C': INCLUDE (copycode used by INCEX2)
*****
* Transferred parameters from INCEX2:
*
* &1& : EMPL-VIEW
* &2& : NAME
* &3& : 'ARCHER'
* &4& : 20
* &5& : 'BAILLET'
*
*
READ (&4&) &1& BY &2& = &3&
  DISPLAY &2&
  IF &2& = &5&
    WRITE 5X 'LAST RECORD FOUND' &2&

```

```

        STOP
    END-IF
END-READ
*
* Statements above will be completed to:
*
* READ (20) EMPL-VIEW BY NAME = 'ARCHER'
*   DISPLAY NAME
*   IF NAME = 'BAILLET'
*     WRITE 5X 'LAST RECORD FOUND' NAME
*     STOP
*   END-IF
* END-READ

```

**Output of Program INCEX2:**

Page 1 05-01-25 16:30:43

```

        NAME
-----
ARCHER
ARCONADA
ARCONADA
ARNOLD
ASTIER
ATHERTON
ATHERTON
ATHERTON
AUBERT
BACHMANN
BAECKER
BAECKER
BAGAZJA
BAILLET
        LAST RECORD FOUND BAILLET

```

**Example 3 - INCLUDE Statement Using Nested Copycodes**

**Program containing INCLUDE statement:**

```

** Example 'INCEX3': INCLUDE (using nested copycodes)
*****
DEFINE DATA LOCAL
1 #A (I4)
END-DEFINE
*
MOVE 123 TO #A
WRITE 'Program INCEX3 ' '=' #A
*
INCLUDE INCEX31C '#A' '5' /* source line is #A := 5
*
*
MOVE 300 TO #A
WRITE 'Program INCEX3 ' '=' #A
*
INCLUDE INCEX32C '''#A''' '''20''' /* source line is #A := 20
*
WRITE 'Program INCEX3 ' '=' #A
END

```

**Copycode INCEX31C to be included:**

```

** Example 'INCEX31C': INCLUDE (copycode used by INCEX3)
*****
* Transferred parameters from INCEX3:
*
* &1& : #A
* &2& : 5
*
*
&1& := &2&
*
WRITE 'Copycode INCEX31C' '=' &1&

```

**Copycode INCEX32C to be included:**

```

** Example 'INCEX32C': INCLUDE (copycode used by INCEX3)
*****
* Transferred parameters from INCEX3:
*
* &1& : '#A'
* &2& : '20'
*
*
WRITE 'Copycode INCEX32C' &1& &2&
*
INCLUDE INCEX31C &1& &2&

```

**Output of Program INCEX3:**

Page 1 05-01-25 16:35:36

```

Program INCEX3 #A: 123
Copycode INCEX31C #A: 5
Program INCEX3 #A: 300
Copycode INCEX32C #A 20
Copycode INCEX31C #A: 20
Program INCEX3 #A: 20

```

**Example 4 - INCLUDE Statement with Concatenated Parameters in Copycode****Program containing INCLUDE statement:**

```

** Example 'INCEX4': INCLUDE (with concatenated parameters in copycode)
*****
DEFINE DATA LOCAL
1 #GROUP
2 ABC(A10) INIT <'1234567890'>
END-DEFINE
*
INCLUDE INCEX4C '#GROUP.' 'ABC' 'AB'
*
END

```

**Copycode INCEX4C to be included:**

```
** Example 'INCEX4C': INCLUDE (copycode used by INCEX4)
*****
* Transferred parameters from INCEX4:
*
* &1& : #GROUP.
* &2& : ABC
* &3& : AB
*
*
WRITE  '=' &2&           /* 'ABC'           results into ABC
WRITE  '=' &1&ABC        /* '#GROUP.' ABC  results into #GROUP.ABC
WRITE  '=' &1&&2&        /* '#GROUP.' 'ABC' results into #GROUP.ABC
WRITE  '=' &1&&3&C       /* '#GROUP.' 'AB' C results into #GROUP.ABC
```

**Output of Program INCEX4:**

Page 1

05-01-25 16:37:59

```
ABC: 1234567890
ABC: 1234567890
ABC: 1234567890
ABC: 1234567890
```